
1. Introduction

1.1. Why do we need to read code? Reading code is executing the program in our head.
It is a necessary skill for understanding programs written by others. It is also important for
comprehending our own code. When trying to solve a problem, we seldom get it right at first try.
So we should ask ’What have I said here?’, that is reading the code just written back. Common
mistake is to be convinced that we wrote exactly what we wanted.

In order to write programs we need to know the basic building blocks very well. The following
exercises all focus on a single aspect of a data structure, a function or a special form. Solving
these will provide the background knowledge in the sense of ‘knowing the tools’.

These exercises alone will not teach how to write code. One has to solve actual problems for
that. Most questions are typical usage examples, but some of them are twisted (if not wicked).

https://egri-nagy.github.io/popbook/

1.2. Instructions. What is the output?, What does the expression evaluate to? The task is to
figure out the ‘meaning’ of the piece of code in the box, which is the result of the computation.
Therefore, there is only a single correct answer, the exact output.

– Each box is a separate question, definitions and symbol bindings are only valid in the
box they appear in.

– The answer is the value of the last expression/line. If the box contains definitions, their
outputs (which can be system dependent) need not be written.

– When the expression leads to an error message, it is enough to state the error, no need
to specify which exception is thrown.

– Indicate the type of the output clearly. For example, sequence operations return sequences
that are lists. When the output is (1 2 3), writing 1 2 3 or [1 2 3] instead are
not acceptable answers. Similarly, when the output is "hello", then writing hello is
incorrect. When the answer is true, writing yes will not earn a mark.

1.3. Recommended method. First, solve these problems without the computer. Write down
the answers, then check them in a REPL by evaluating the code snippets. If your answer is
different, then check your notes or textbook or documentation for the given function and data
structure. Simply copy-pasting the expressions into the REPL without thinking achieves very
little learning. This is also the reason for not giving the solutions in this document.

Here are some hints for building up reading skills. It makes sense to keep asking these
questions.

– Data literals (e.g. numbers, strings, characters) are easy to read. They just evaluate to
themselves. Their meaning is immediate.

– For a symbol, we need to find what does it bind to? What is the meaning? Is it a
function? Is it a data item? We need to go backwards to find where the symbol gets
defined. Is it in a let statement? Is it defined in the environment? Is it an argument to
a function?

– If the expression is composite, work from the inside out. Evaluate the sub-expressions
first. Make notes. Rewrite the full expression by substituting the values for the sub-
expressions.

– If a function is called, what are its input arguments? Does the function expect a number
of arguments? or a single collection?

2. Function composition first

10

−→ 10

(identity 10)

−→ 10

(inc 2)

−→ 3

(dec 9)

−→ 8

(inc (dec 0))

−→ 0

((comp inc dec) 0)

−→ 0

(dec (dec 0))

−→ -2

((comp dec dec dec) 0)

−→ -3

((comp (comp dec dec) (comp dec dec)) 5)

−→ 1

((comp identity identity) 1)

−→ 1

((comp identity inc) 1)

−→ 2

(inc 11.11)

−→ 12.11

(dec 0.01)

−→ -0.99

((comp inc) 1)

−→ 2

((comp) 42)

−→ 42

(((comp comp comp) dec dec) 1)

−→ -1

3. Arithmetic done with functions

(+ (- 6 4) (/ 6 3))

−→ 4

(+ 1 (* 2 (- 1 (/ 12 4))))

−→ -3

(+)

−→ 0

(-)

−→ ArityException Wrong number of args (0) passed to: core/-

(*)

−→ 1

(/)

−→ ArityException Wrong number of args (0) passed to: core//

(+ 3)

−→ 3

(* 5)

−→ 5

(/ 3)

−→ 1/3

(- 5)

−→ -5

(+ (* 5 2) (- 4 3) (*))

−→ 12

((comp - *) 2 3 4)

−→ -24

((comp / *) 2 3)

−→ 1/6

(* (inc 1) (inc (inc 1)))

−→ 6

(/ 10 4)

−→ 5/2

(/ 10.0 4)

−→ 2.5

(/ 10 4.0)

−→ 2.5

4. Asking yes-or-no questions: predicates

(zero? 0)

−→ true

(zero? 1)

−→ false

(pos? 1)

−→ true

(pos? 1111)

−→ true

(pos? 0)

−→ false

(neg? 0)

−→ false

(neg? -2)

−→ true

(= (+ 1 2 3) (* 1 2 3))

−→ true

(<= 2 2 2)

−→ true

(<= 2 1 2)

−→ false

(< 2 1 2)

−→ false

(< 2 2 2)

−→ false

(< 2 3 4)

−→ true

(> 2 3)

−→ false

(> 2 2)

−→ false

(>= 2 2)

−→ true

(fn? +)

−→ true

(fn? -)

−→ true

(fn? identity)

−→ true

(fn? (+))

−→ false

(number? (+))

−→ true

(rational? (/ 7 3))

−→ true

(rational? 2)

−→ true

(float? 2)

−→ false

(float? 2.0)

−→ true

(integer? (+ 1 2 3 4 5))

−→ true

(integer? (+ 1 2 3.0 4 5))

−→ false

(float? (+ 1 2 3.0 4 5))

−→ true

(number? 12.1)

−→ true

(number? 0)

−→ true

(number? (/ 1 19))

−→ true

5. Strings

(char? \x)

−→ true

(char? \space)

−→ true

(char? \8)

−→ true

(= 9 \9)

−→ false

(string? "Granny Weatherwax")

−→ true

(string? " ")

−→ true

(string? "")

−→ true

(string? "12")

−→ true

(number? "12")

−→ false

(= \space " ")

−→ false

(= (str \space) " ")

−→ true

(str \1 2 "3" (- 5 1))

−→ "1234"

(str "The answer:" 42)

−→ "The answer:42"

(str "The answer: " 42)

−→ "The answer: 42"

(str "The answer: " 42 ".")

−→ "The answer: 42."

(str "The answer: " (* 6 7) ".")

−→ "The answer: 42."

WARNING! The following questions assume that the string library is loaded by
(require ’[clojure.string :as string]).

(string/upper-case "helloooo!")

−→ "HELLOOOO!"

(string/capitalize (string/lower-case "HELLO!"))

−→ "Hello!"

(string/ends-with? "mango" "go")

−→ true

(string/ends-with? "mango" "GO")

−→ false

(string/starts-with? "How to solve it?" "?")

−→ false

(string/replace "banana" "a" "e")

−→ "benene"

(string/replace (string/replace "banana" "a" "-") "-n" "x")

−→ "bxx-"

6. List, the most fundamental collection

'(1 2 3)

−→ (1 2 3)

(1 2 3)

−→ ClassCastException java.lang.Long cannot be cast to clojure.lang.IFn

(list 4 5)

−→ (4 5)

(list 4 5 '(6 7))

−→ (4 5 (6 7))

(list 4 5 '(6 7 (8 9)))

−→ (4 5 (6 7 (8 9)))

(cons 11 '(19 13))

−→ (11 19 13)

(cons 7 (cons 5 (list 3 4)))

−→ (7 5 3 4)

(first '(8 3 5))

−→ 8

(first '())

−→ nil

(rest (5 3 8))

−→ ClassCastException java.lang.Long cannot be cast to clojure.lang.IFn

(rest '(5 3 8))

−→ (3 8)

(rest '())

−→ ()

((comp rest rest) '(2 3 4))

−→ (4)

((comp first rest) '(2 3 4))

−→ 3

((comp first first) '(2 3 4))

−→ IllegalArgumentException Don’t know how to create ISeq from: java.lang.Long

(last '(9 3 2))

−→ 2

(last '())

−→ nil

(reverse '(1 2 3))

−→ (3 2 1)

((comp first reverse) '(1 2 3))

−→ 3

(empty? ())

−→ true

(empty? '(1))

−→ false

(count ())

−→ 0

(count (list \a \b))

−→ 2

(concat '(1) '(3) '(2))

−→ (1 3 2)

(concat)

−→ ()

7. Vectors

(vector \a \b \c)

−→ [\a \b \c]

(vector [1 2])

−→ [[1 2]]

(vec [1 2])

−→ [1 2]

(vec '(\a \b))

−→ [\a \b]

(vec [])

−→ []

(vector [])

−→ [[]]

(count [1 "two" 3])

−→ 3

(count [])

−→ 0

(count [[]])

−→ 1

(count [[[]]])

−→ 1

(count [[[[[[[]]]]]]])

−→ 1

(count [[][][]])

−→ 3

(nth [\a \b \c] 0)

−→ \a

(nth [\a \b \c] 2)

−→ \c

(nth [\a \b \c] 3)

−→ IndexOutOfBoundsException

([\a \b \c] 0)

−→ \a

([\a \b \c] 1)

−→ \b

([\a \b \c] 3)

−→ IndexOutOfBoundsException

(nth [] 0)

−→ IndexOutOfBoundsException

(conj [] 1)

−→ [1]

(conj [] 1 2)

−→ [1 2]

(conj [] 1 2 3)

−→ [1 2 3]

(conj [\a \b] 3)

−→ [\a \b 3]

(conj [] [])

−→ [[]]

(conj [] [] [])

−→ [[] []]

(conj [1 2] [3])

−→ [1 2 [3]]

(vec (range 3))

−→ [0 1 2]

(vector (range 3))

−→ [(0 1 2)]

((comp ["hello" "world" "!"] [2 1 0]) 2)

−→ "hello"

8. Making memories

(def a 2)
(* 3 a)

−→ 6

(def a 5)
(def b 7)
(* a b)

−→ 35

(= 'x "x")

−→ false

(def x "x")
(= "x" x)

−→ true

(def x "grape")
(def x "apple")
x

−→ "apple"

(let [x 2] (* 3 x))

−→ 6

(let [x 2, y 7] (* y x))

−→ 14

(let [s "world"] (str "Hello " s "!"))

−→ "Hello world!"

(let [s "everyone"] "Hello world!")

−→ "Hello world!"

(def x 2)
(let [x 100] (inc x))

−→ 101

(def x 2)
(let [x 100] (inc x))
x

−→ 2

(let [x 0 y (inc x) z (dec y)] [x y z])

−→ [0 1 0]

9. Defining functions

((fn [x] (+ x 2)) 0)

−→ 2

((fn [x] (+ x 2)) 1)

−→ 3

((fn [x y] (+ x (* 2 y))) 1 2)

−→ 5

((fn [x] (+ x (* 2 y))) 1 2)

−→ Unable to resolve symbol: y in this context

(def x 31)
((fn [x] (+ x 2)) 1)

−→ 3

(defn f [s] (str "Input was " s "."))
(f 42)

−→ "Input was 42."

(defn f [n] (* 3 n))
((comp inc f) 1)

−→ 4

10. Manual function calls

(+ [1 2 3])

−→ Cannot cast clojure.lang.PersistentVector to java.lang.Number

(apply + [1 2 3])

−→ 6

(max 5 2 7 4)

−→ 7

(max [5 2 7 4])

−→ [5 2 7 4]

(apply max [5 2 7 4])

−→ 7

(apply str [\h \i])

−→ "hi"

11. Lazy lists of numbers

(range 2)

−→ (0 1)

(range 1 2)

−→ (1)

(range 1 2 3)

−→ (1)

(range 2 1 3)

−→ ()

(range 0 1 0.4)

−→ (0 0.4 0.8)

(take 2 (drop 2 (range)))

−→ (2 3)

(take 3 '(3 3 3 3))

−→ (3 3 3)

(take-while neg? (range -2 2))

−→ (-2 -1)

(drop-while neg? (range -2 2))

−→ (0 1)

(take-while zero? '(-1 0 1))

−→ ()

(drop-while zero? '(-1 0 1))

−→ (-1 0 1)

(drop-while zero? '(-1 0 1 -1))

−→ (-1 0 1 -1)

(drop-while neg? '(-1 0 1 -1))

−→ (0 1 -1)

(count (take 5 (range)))

−→ 5

12. Functional collection transformation

(map inc [1 2 3 4 5])

−→ (2 3 4 5 6)

(map dec [1 2 3 4 5])

−→ (0 1 2 3 4)

(map range [2 3])

−→ ((0 1) (0 1 2))

(mapcat range [2 3])

−→ (0 1 0 1 2)

(map str [1 2 3])

−→ ("1" "2" "3")

(map reverse [[1 2] [3 4]])

−→ ((2 1) (4 3))

(map reverse (map range [1 2 3]))

−→ ((0) (1 0) (2 1 0))

(map even? (range 5))

−→ (true false true false true)

(map odd? [0 1 2 3])

−→ (false true false true)

(map str [1 \a "2" ()])

−→ ("1" "a" "2" "()")

13. Selections

(filter even? [10 11 12 13])

−→ (10 12)

(filter odd? [10 11 12 13])

−→ (11 13)

(filter even? [1 3 5 7])

−→ ()

(filter number? [1 "2" \3])

−→ (1)

(filter string? [1 "2" \3])

−→ ("2")

(filter char? [1 "2" \3])

−→ (\3)

(remove nil? [[] nil () 0])

−→ ([] () 0)

(remove even? (range 6))

−→ (1 3 5)

(filter char? [1 \a 2 \b])

−→ (\a \b)

14. Conditionals

(if true 42 24)

−→ 42

(if false 42 24)

−→ 24

(let [x 2] (cond (string? x) "just a word"
(= x 2) "two"
true "not two"))

−→ "two"

(let [x 4] (cond (string? x) "just a word"
(= x 2) "two"
true "not two"))

−→ "not two"

(let [x "2"] (cond (string? x) "just a word"
(= x 2) "two"
true "not two"))

−→ "just a word"

(not nil)

−→ false

(not 42)

−→ true

(and false nil)

−→ false

(or false nil)

−→ nil

(and 1 2)

−→ 2

(or 1 2)

−→ 1

15. Reduce

(reduce conj [9] [2 3] )

−→ [9 2 3]

(reductions conj [9] [2 3] )

−→ ([9] [9 2] [9 2 3])

(into [9] [2 3] )

−→ [9 2 3]

(into '(1 2) '(3 4 5))

−→ (5 4 3 1 2)

(reductions conj '(1 2) '(3 4 5))

−→ ((1 2) (3 1 2) (4 3 1 2) (5 4 3 1 2))

(reduce max 0 [2 5 3])

−→ 5

(reduce max 6 [2 5 3])

−→ 6

(reductions max 0 [2 5 3])

−→ (0 2 5 5)

(reductions max 6 [2 5 3])

−→ (6 6 6 6)

(defn rf
[v x]
(if (odd? x)

(conj v x)
(conj v (dec x))))

(reduce rf [] [1 2 3 4 5])

−→ [1 1 3 3 5]

(defn rf [v x] (if (even? x) (conj v x) (conj v (inc x))))
(reduce rf [] [1 2 3 4 5])

−→ [2 2 4 4 6]

(reduce * [1 2 3])

−→ 6

(reduce + [1 3])

−→ 4

16. Hash-map

(hash-map \a \b \b \a)

−→ {\a \b, \b \a}

(hash-map :x 10 :y 11)

−→ {:y 11, :x 10}

({1 2 3 4} 3)

−→ 4

({1 2 3 4} 2)

−→ nil

(zipmap (range 4) (reverse (range 4)))

−→ {0 3, 1 2, 2 1, 3 0}

(hash-map :a 1 :b 2 :c)

−→ IllegalArgumentException No value supplied for key: :c

(:a {:a "1" :b 2})

−→ "1"

(1 {1 2 3 4})

−→ ClassCastException java.lang.Long cannot be cast to clojure.lang.IFn

({:a 1 :b 2} :c)

−→ nil

({:a 1 :b 2} :c :nothing)

−→ :nothing

(get {:a 1 :b 2} :c)

−→ nil

(get {:a 1 :b 2} :c "No such key!")

−→ "No such key!"

(map {1 "one" 2 "two"} [0 1 2 3])

−→ (nil "one" "two" nil)

(assoc {:a 3} :b 2)

−→ {:a 3, :b 2}

(keys {:title "Solaris" :author "Stanislaw Lem" :year 1961})

−→ (:title :author :year)

(vals {:title "Solaris" :author "Stanislaw Lem" :year 1961})

−→ ("Solaris" "Stanislaw Lem" 1961)

17. Hash-sets

(hash-set "hello" 17 \c)

−→ #{\c "hello" 17}

(set [81 72])

−→ #{72 81}

(hash-set [6 28 496])

−→ #{[6 28 496]}

(apply hash-set [6 28 496])

−→ #{496 6 28}

(set [6 28 496])

−→ #{496 6 28}

(sorted-set "helloooooo!!")

−→ #{"helloooooo!!"}

(apply sorted-set "helloooooo")

−→ #{\e \h \l \o}

18. Sequence abstraction

(seq "coffee")

−→ (\c \o \f \f \e \e)

(seq (zipmap (range 4) [\a \b \c \d]))

−→ ([0 \a] [1 \b] [2 \c] [3 \d])

(seq {:a 3 :b 2})

−→ ([:a 3] [:b 2])

(seq [1 2 3 4])

−→ (1 2 3 4)

(count {:a 3 :b 4})

−→ 2

19. Iteration

(take 4 (iterate inc 2))

−→ (2 3 4 5)

(take 4 (iterate dec 0))

−→ (0 -1 -2 -3)

(take 8 (iterate (fn [x] (* 2 x)) 1))

−→ (1 2 4 8 16 32 64 128)

(take 4 (iterate (fn [v] (conj v 1)) []))

−→ ([] [1] [1 1] [1 1 1])

